

Flash Tools for Developers: Planar Graphers in ActionScript 3.0

This paper is a companion to the online article at the MathDL Digital Classroom Resources “Flash Tools for Developers (AS3): Graphing Curves on the Plane” by Barbara Kaskosz and Doug Ensley. This paper provides a brief introduction to ActionScript 3.0, a note on the templates presented in the article, and complete documentation of the custom classes in the package bkde.as3. provided in the article.*

ActionScript 3.0 versus ActionScript 2.0

The new version of Flash, Flash CS3, released by Adobe in April of 2007, allows you to continue development in ActionScript 2.0 (AS2). It also gives you an option of writing your code in ActionScript 3.0 (AS3), the new version of the language. Although AS2 is somewhat easier than AS3, we recommend that you switch to AS3. The gains in speed, performance, and elegance are enormous. AS3 is a true object-oriented language and it is up to ten times faster than the older version.

The changes in AS3 in comparison with AS2 vary from minor changes in syntax (for example: `TextField.visible` versus `TextField._visible`) to deep changes in the display model and event handling model. There are two new visual classes in AS3, `Sprite` and `Shape`, in addition to the `MovieClip` class. Creating subclasses of each of the three classes is intuitive and easy. Hence, creation of custom visual classes, like the `GraphingBoard` class contained in our article, became much easier and more natural.

To learn AS3, we recommend documentation on the Adobe site:

<http://www.adobe.com/support/documentation/en/flash/>

as well as help files that come with Flash CS3.

Books on AS3 are only now coming out or are scheduled to come out. We recommend freshly from the printing presses:

Essential ActionScript 3.0 by C. Moock, O'Reilly, 2007.

Due in September and October are:

ActionScript 3 Bible by R. Braunstein, M. Wright, J. Lott, and J. Noble, Wiley, 2007.

Adobe Flash CS3 Professional Bible by R. Reinhardt and S. Dowd, Wiley, 2007.

Soon, we will post an extensive set of tutorials in AS3 at the MathDL Flash Forum Learning Center at:

<http://www.math.uri.edu/~flashcenter/>

The Templates

The article provides three extensively commented templates: two templates for building a function grapher and one for building a parametric curves grapher. The templates are similar in functionality to the templates in our earlier Flash Tools for Developers articles, except that they are written in AS3. If you are migrating from AS2 to AS3, comparing the two versions of each template should be beneficial.

The GraphingBoard class that is behind each template has been redone thoroughly. It extends Sprite now and it has much new functionality. Below, you will find complete documentation of all the custom classes in the package `bkde.as3.*`. In our upcoming articles, we will be adding new classes to the package.

Classes in the Package `bkde.as3.*`

Note: All methods and properties of the classes listed below are instance methods and properties. None of the classes has static methods or properties.

`bkde.as3.parsers.CompiledObject`

Description

`CompiledObject` is a helper class for `MathParser`. It creates a convenient datatype to be returned by `doCompile` method of `MathParser`. This datatype is an object with three properties listed below which comprise the results of compiling a mathematical formula into a form suitable for evaluation.

Constructor

The constructor is evoked by the keyword `"new"` and takes no parameters:

- `new CompiledObject();`

You shouldn't encounter the need to use the constructor since the only instances of `CompiledObject` that can conceivably be useful are those returned by `doCompile` method of `MathParser`.

Public Methods

None.

Public Properties

CompiledObject has three public instance properties.

- **`instance.PolishArray : Array`**

When the instance is returned by doCompile method of MathParser, the array represents a mathematical formula in the Polish notation.

- **`instance.errorStatus : Number`**

When the instance is returned by doCompile method of MathParser, the property has value 1 if an error is found and 0 otherwise.

- **`instance.errorMes : String`**

When the instance is returned by doCompile method of MathParser, the string contains a message to the user indicating where in the input a mistake was found.

bkde.as3.parsers.MathParser

Description

An instance of MathParser (you create an instance using the class's constructor described below) will compile a string that represents a mathematical formula (usually the user's input) and then calculate the values of the compiled formula for given values of variables that are recognized by the instance. "Compiling" consists of rewriting a formula in a form suitable for evaluation; that is, in the Polish notation. Compiling will be successful if the user obeys by the simple syntax rules described at the end of this section.

Constructor

The constructor is evoked with the word "new":

- **`new MathParser(parameter1:Array)`**

The constructor takes one parameter which is an array. For the MathParser's instance to do what you want it to do, the parameter has to be an array of strings. The strings represent the names of variables that the instance will recognize. For example:

```
var procFun:MathParser = new MathParser(["x", "y"]);
```

The instance "procFun" will recognize the variables x and y.

```
var procFormula:MathParser = new MathParser(["t"]);
```

The instance "procFormula" will recognize `t` as a variable. MathParser knows the constants `e` and `pi`. Do not enter them into the constructor.

Note: Variables have to be entered as strings. It is:

```
procFormula:MathParser = new MathParser(["x"]);
```

and not:

```
procFormula:MathParser = new MathParser([x]);
```

Variables can be comprised of more than one letter, e.g.:

```
procFormula:MathParser = new MathParser(["tension"]);
```

It is important to remember the order in which you pass your variables to the constructor since the evaluator method of the parser will expect values for those variables in the same order.

Public Methods

MathParser has two public instance methods.

- ***instance.doCompile(parameter1:String): CompiledObject***

The method takes a string (typically a mathematical formula entered by the user) and returns an instance of CompiledObject. If no mistakes in syntax are found, the PolishArray property of the returned CompiledObject instance represents the formula in the Polish notation, `errorStatus=0`, `errorMes=""`. If a mistake is found, `errorStatus=1`, `errorMes` contains a message indicating where the mistake was found, `PolishArray=[]`;

- ***instance.doEval(parameter1:Array,parameter2:Array): Number***

The method takes two parameters, both arrays. For the method to be useful, the first array must be the PolishArray property of a CompiledObject returned by doCompile method. The second parameter represents an array of numerical values for the variables recognized by the instance of MathParser. (The same variables that you passed to the constructor of your MathParser instance.) Under these conditions, doEval will return the value of the formula represented by the PolishArray for the specified values of the variables.

Public Properties

None.

Examples of Use

In our *Function Grapher in ActionScript 3.0 – Template 1*, we use MathParser as follows:

```
var procFun:MathParser = new MathParser(["x"]);
```

We created an instance that will recognize x as a variable. The user enters text into the input text field, InputBox1, on the Stage. The text entered represents a formula for the first function to be graphed. Within the function makeGraphs, we retrieve the user's input and store in a string variable sFunction1; we declare other variables related to parsing:

```
var sFunction1:String= "";
.....
var compObj1:CompiledObject;
.....

sFunction1=InputBox1.text;
```

If the input is not empty, we apply to it doCompile method of the parser and store the result in the variable compObj1. (The method returns an instance of CompiledObject.)

```
compObj1=procFun.doCompile(sFunction1);
```

If compObj1.errorStatus=0, the user entered a formula properly. The formula is now stored in the form ready for evaluation (that is, rewritten in the Polish notation) in compObj1.PolishArray. To evaluate the formula for a given value of x, say xmin which represents a number defined previously in the script, we use doEval method of the parser as follows:

```
procFun.doEval(compObj1.PolishArray, [xmin]);
```

In makeGraphs function the latter value as well as the values of the formula for other values of x between xmin and xmax are stored as elements of an array, flArray.

Syntax Accepted by MathParser

The parser expects calculator-like syntax: e.g.:

$$\sin(2*x^2)-e^{-x}+\tan(\pi*x)/2$$

Multiplication must be entered as *. Arguments of functions must be enclosed in parentheses. The parser is case-insensitive and blind to white spaces. It recognizes the constants e and pi. Here is the list of functions that the parser knows:

sin(·), cos(·), tan(·), asin(·), acos(·), atan(·), ln(·), sqrt(·), abs(·), ceil(·), floor(·), round(·), max(·,·), min(·,·).

Addition, multiplication, division and exponentiation are denoted by the usual symbols +, *, /, ^, subtraction and unary minus by -.

bkde.as3.parsers.RangeObject

Description

RangeObject is a helper class for RangeParser. It creates a convenient datatype to be returned by parseRangeTwo and parseRangeFour methods of RangeParser. This datatype is an object with three properties listed below which comprise the results of compiling the user's range input for two variables, say x and y, or for one variable, say a parameter t. In most of our applets the range boxes allow for numerical entries as well as for entries containing pi, like pi/4, 2*pi etc.. Such entries have to be parsed and checked for validity.

Constructor

The constructor is evoked by the keyword "new" and takes no parameters:

- **new RangeObject();**

You shouldn't encounter the need to use the constructor since the only instances of RangeObject that can conceivably be useful are those returned by parseRangeTwo or parseRangeFour methods of RangeParser.

Public Methods

None.

Public Properties

RangeObject has three public instance properties.

- **instance.Values : Array**

When the instance is returned by one of the methods of RangeParser, the array represents four range values (if ranges for two variables are being parsed) or two range values if the range for one variable is being parsed.

- **instance.errorStatus : Number**

When the instance is returned by one of the RangeParser methods, the property has value 1 if an error is found and 0 otherwise.

- **instance.errorMes : String**

When the instance is returned by one of the RangeParser methods, the string contains a message to the user indicating where in the input a mistake was found.

bkde.as3.parsers.RangeParser

Description

In most of our applets the range boxes for x and y or for a parameter, say t, allow for numerical entries as well as for entries containing pi, like pi/4, 2*pi etc.. Such entries have to be parsed and checked for validity.

Constructor

The constructor is evoked with the word “new” and takes no parameters:

- **new RangeParser();**

For example

```
var procRange:RangeParser = new RangeParser();
```

Public Methods

RangeParser has two public instance methods.

- **instance.parseRangeTwo(parameter1:String, parameter2:String):RangeObject**

The method takes two strings (typically the user’s entries in range boxes for a parameter t, for example) and returns an instance of RangeObject. If the entries are found to be valid numerical entries (or valid entries containing pi), and the first entry is less than the second entry, the Values property of the returned RangeObject contains the two range values. In that case, errorStatus=0, errorMes= “”. If a mistake was found, errorStatus=1, errorMes contains a message indicating where the mistake was found, Values=[];

- **instance.parseRangeFour(parameter1:String, parameter2:String, parameter3:String, parameter4:String):RangeObject**

The method takes four strings (typically the user’s entries for x and y ranges) and returns an instance of RangeObject. If the entries are found to be valid numerical entries (or valid entries containing pi), the first entry is less than the second entry, the third entry is less than the fourth entry, then the Values property of the returned RangeObject contains the corresponding four range values. In that case, errorStatus=0, errorMes= “”. If a mistake was found, errorStatus=1, errorMes contains a message indicating where the mistake was found, Values=[];

Public Properties

None.

Examples of Use

In our *Function Grapher in ActionScript 3.0 – Template 1*, we use RangeParser as follows:

```
var procRange:RangeParser = new RangeParser();  
.....
```

Then inside the function makeGraphs:

```
sXmin=XminBox.text;  
  
sXmax=XmaxBox.text;  
  
sYmin=YminBox.text;  
  
sYmax=YmaxBox.text;  
  
oRange=procRange.parseRangeFour(sXmin,sXmax,sYmin,sYmax);  
  
if(oRange.errorStatus==1){  
  
    board.ErrorBox.visible=true;  
  
    board.ErrorBox.text="Error. "+oRange.errorMes;  
  
    return;  
  
}  
  
xmin=oRange.Values[0];  
  
xmax=oRange.Values[1];  
  
ymin=oRange.Values[2];  
  
ymax=oRange.Values[3];
```

“board” above is an instance of GraphingBoard which controls the error display box board.ErrorBox.

bkde.as3.boards.GraphingBoard

Description

GraphingBoard is the main visual class for creating customizable planar graphers: function graphers, parametric curves graphers, etc.. An instance of GraphingBoard draws a rectangular graphing board (at runtime), the vertical and horizontal coordinate axes as well as graphs of functions or parametric curves. Any instance of GraphingBoard contains and controls an error display text field where messages to the user can be displayed. It also contains a coordinate

display text field in which the values of the horizontal and the vertical coordinates are displayed when the user mouses over the graphing board. An instance of GraphingBoard can enable the user to draw within the graphing board with the mouse.

The layout, the colors, and the sizes of all elements of an instance of GraphingBoard are all easily customizable via instance methods of the class.

An instance of GraphingBoard sets x and y ranges in functional terms (usually based on the user's input), and provides public methods for translating pixel coordinates of a point into its functional coordinates and vice versa. For simplicity of this presentation, we will assume most of the time that the horizontal and vertical variables in your applet are named "x" and "y" although you may choose different names using the method "enableCoordsDisp(..., ...)" described later. The pixel coordinates in the Flash's coordinate system are, however, always x and y.

GraphingBoard extends Sprite. Thus, it inherits from Sprite. In particular, you can control the position of your instance of GraphingBoard within the main movie with the Sprite methods and properties:

instance.x

instance.y

These properties set the x and the y coordinates in pixels of the upper left corner of your instance of GraphingBoard with respect to the upper left corner of the parent movie. Recall that in Flash, the x coordinate increases to the right, the y coordinate increases as you go down.

In our *Function Grapher in ActionScript 3.0 – Template 1*, we have an instance of GraphingBoard named "board", a child of the main movie. We position "board" within the main movie with:

```
board.x=20;
```

```
board.y=20;
```

Constructor

The constructor is evoked with the word "new" and takes two numerical parameters. The parameters are the width and the height, in pixels, of the rectangular graphing board which will be drawn:

▪ **`new GraphingBoard(w:Number,h:Number) ;`**

In our *Function Grapher in ActionScript 3.0 – Template 1*, we create a 350 by 350 graphing board as follows:

```
var board:GraphingBoard = new GraphingBoard(350,350);
```

We store our instance of GraphingBoard in a variable called “board” whose datatype is GraphingBoard. We pass to the constructor the size of our graphing board; all other attributes will be set using the methods of the class.

Public Methods – Graphing Board Appearance

- **`instance.changeBackColor(h:Number): void`**

The method controls the background color of the graphing board created by the instance. The numerical parameter should be the desired color in the hexadecimal form.

Default: white.

To continue with examples from *Function Grapher in ActionScript 3.0 – Template 1* where the instance of GraphingBoard is stored in “board”:

```
board.changeBackColor(0x000000);
```

We set the background to black.

- **`instance.changeBorderColorAndThick(h:Number,t:Number): void`**

The method controls the color and the thickness of the border of a graphing board created by the instance. The first parameter passed to the method should be the hexadecimal form for the desired color.

Default: black, 1.

In *Function Grapher in ActionScript 3.0 – Template 1* where the instance of GraphingBoard is stored in “board” we use:

```
board.changeBorderColor(0xFFFFFF,1);
```

We set the border color to white, its thickness to 1.

Public Methods – Coordinate Axes Appearance

- **`instance.setAxesColorAndThick(h:Number,t:Number): void`**

The method sets the color and the thickness, in pixels, of the x and y axes. The color should be passed to the method in its hexadecimal form.

Default: black, 0.

(Thickness set to 0 produces a line 1 pixel wide whose thickness will not change with rescaling.)

In *Function Grapher in ActionScript 3.0 – Template 1*, we have:

```
board.setAxesColorAndThick(0xCCCCCC,0);
```

We set the color of the coordinate axes to light gray, their thickness to 0.

Note: Colors and thickness of graphs of functions or curves are passed directly to the graphing method of GraphingBoard as shown later in this guide. That way those attributes can vary from graph to graph.

Public Methods – Drawing by the User

- **`instance.enableUserDraw(h:Number,t:Number): void`**

The method enables the user to draw on the graphing board with the mouse. It sets the color and the thickness, in pixels, of the user's drawing.

Default: enabled in red with thickness 0.

In *Function Grapher in ActionScript 3.0 – Template 1*, we have:

```
board.enableUserDraw(0xFFFF00,1);
```

We enable the user to draw in yellow with thickness 1.

If you want to disable the drawing capability, use the method:

- **`instance.disableUserDraw(): void`**

Public Methods – Error Display

An instance of GraphingBoard controls the text field for displaying error messages to the user. You can control the appearance and the position of the error text field with the following methods.

- **`instance.setErrorBoxSizeAndPos(w:Number,h:Number,xpos:Number,ypos:Number): void`**

The parameters determine: the width, the height (in pixels) of the error text field, and its x and y position relative to your instance of GraphingBoard.

Default: The text field is positioned over the upper half of the graphing board created by the instance.

In *Function Grapher in ActionScript 3.0 – Template 1*, we have:

```
board.setErrorBoxSizeAndPos(310,120,20,20);
```

You can set visual attributes of the error box with the method:

- **`instance.setErrorBoxFormat (c1:Number,c1:Number,c3:Number,s:Number) :`**
`void`

The parameters determine: the background color, the border color, the text color, and the text size. All colors should be passed in hex.

Default values: white, white, black, 12.

In *Function Grapher in ActionScript 3.0 – Template 1*, we have:

```
board.setErrorBoxFormat (0x000000,0x000000,0xCCCCCC,12);
```

We chose black background, black border, and gray text of size 12.

Note: The error display text field is a public property of GraphingBoard:

- **`instance.ErrorBox`**

Hence, you can control its attributes directly through methods of Flash's TextField class. It is easier, however, to use the methods of GraphingBoard to set properties of the error box.

We made ErrorBox property public to give you easy control over the visibility of ErrorBox and the text displayed in it. (The error box is visible when the user made an error; its text is an error message to the user determined by the kind of error found.)

Note: The initial visibility of the error box is set to false.

In *Function Grapher in ActionScript 3.0 – Template 1*, we have the following code within makeGraphs function which displays an error message to the user if a mistake in range entries is found:

```
sXmin=XminBox.text;

sXmax=XmaxBox.text;

sYmin=YminBox.text;

sYmax=YmaxBox.text;

oRange=procRange.parseRangeFour (sXmin,sXmax,sYmin,sYmax);

if (oRange.errorStatus==1) {

    board.ErrorBox.visible=true;

    board.ErrorBox.text="Error. "+oRange.errorMes;

    return;

}
```

Public Methods – Coordinates Display

An instance of GraphingBoard controls the text field for displaying the values of the horizontal and vertical variables (in functional terms) when the user mouses over the graphing board. You can control the appearance and the position of the coordinate text field with the following methods.

- **`instance.setCoordsBoxSizeAndPos (w:Number, h:Number, xpos:Number, ypos:Number) :void`**

The parameters determine: the width, the height (in pixels) of the coordinate text field, and its x and y position in pixels relative to your instance of GraphingBoard.

Default: The text field is positioned in the lower left corner a graphing board created by the instance.

In *Function Grapher in ActionScript 3.0 – Template 1*, you can see:

```
board.setCoordsBoxSizeAndPos (60,40,20,300);
```

You can set visual attributes of the coordinate box with the method:

- **`instance.setCoordsBoxFormat (c1:Number, c1:Number, c3:Number, s:Number) :void`**

The parameters determine: the background color, the border color, the text color, and the text size. All colors should be passed in hex.

Default values: white, white, black, 12.

In *Function Grapher in ActionScript 3.0 – Template 1*, we use the method:

```
board.setCoordsBoxFormat (0x000000,0x000000,0xCCCCCC,12);
```

We chose black background, black border, and gray text of size 12.

You can disable the coordinate display box using the method:

- **`instance.disableCoordsDisp() : void`**

Default: enabled.

You can use the “enableCoordsDisp” method to enable coordinates display and to change the names of your horizontal and vertical variables displayed in the coordinates display box from the default “x” and “y” :

- **`instance.enableCoordsDisp(h:String,v:String): void`**

The coordinates displayed in the box will be named according to the string parameters `h` and `v` that you pass to the method.

Default: enabled with coordinates named “x” and “y”.

We do not use the latter method in our templates. If you want your coordinates to be named, for example, “t” and “s”, and your instance of `GraphingBoard` is “board” you call:

```
board.enableCoordsDisp("t","s");
```

Public Methods – Tracing Cursors

`GraphingBoard` provides methods helpful for building graph-tracing mechanisms into your applet. The class allows two styles for the tracing cursor: “cross” and “arrow”. The corresponding method is

- **`instance.setTraceStyle(s:String): void`**

The string parameter “s” has two values that will produce a cursor: “cross” and “arrow”.

Default value: “cross”.

In our template *Parametric Curves in ActionScript 3.0 – Basic*, we chose the arrow:

```
board.setTraceStyle("arrow");
```

For each of the two styles you have much control over the size and the color of the tracing cursor. Below are the corresponding methods.

- **`instance.setCrossSizeAndThick(s:Number,t:Number): void`**

The method sets the size and the line thickness for the cross cursor.

Default: 6, 1.

In *Function Grapher in ActionScript 3.0 – Template 2*, you see:

```
board.setCrossSizeAndThick(6,2);
```

We wanted the cursor thicker than the default value.

- **`instance.setCrossColor(c:Number): void`**

The method sets the color for the cross cursor. The color should be passed in its hex form.

Default: black.

In *Function Grapher in ActionScript 3.0 – Template 2*, we have:

```
board.setCrossColor(0x9900FF);
```

We chose magenta for our cursor.

- **`instance.setCrossPos(s:Number,t:Number): void`**

The method sets the position of the cross cursor, in pixels, relative to the instance's upper left corner.

Default: The upper left corner.

- **`instance.crossVisible(b:Boolean): void`**

The method sets the visibility of the cross cursor to “false” or “true” as in:

```
board.crossVisible(true);
```

Default: false.

The two last methods are used to move the cursor along a given graph and make it visible or invisible depending on the user's actions. You will find tracing functionality in *Function Grapher in ActionScript 3.0 – Template 2*.

- **`instance.getCrossSize(): Number`**

The method returns cross' size in case you need it for positioning purposes. We have similar methods for the arrow cursor.

- **`instance.setArrowSize(s:Number): void`**

The method sets the size for the arrow cursor.

Default: 10.

- **`instance.setArrowColor(c:Number): void`**

The method sets the color for the arrow cursor. The color should be passed in its hex form.

Default: black.

- **`instance.setArrowPos(s:Number,t:Number,r:Number): void`**

The method sets the x and y coordinates, in pixels, of the arrow cursor relative to the upper left corner an instance, $x=s$, $y=t$. The last parameter represents the rotation of the arrow, counterclockwise, in degrees. The rotation parameter allows the arrow to move along a curve and rotate accordingly.

Default: The upper left corner, pointing upwards.

The last method is used to build a tracing mechanism in which an arrow traces a parametric curve when the user slides a slider. You will find an example of such mechanism in *Parametric Curves in ActionScript 3.0 – Basic*.

- **`instance.arrowVisible(b:Boolean): void`**

The method sets the visibility of the arrow cursor to “false” or “true”.

Default: false.

- **`instance.getArrowSize(): Number`**

The method returns arrow’s size in case you need it for positioning purposes.

Public Methods – Graphing

For every instance of GraphingBoard the maximum number of graphs that can be displayed simultaneously should be set via the method:

- **`instance.setMaxNumGraphs(a:int): void`**

Default: 5.

For example, in *Function Grapher in ActionScript 3.0 – Template 1*, we do want to graph three functions so we don’t have to call the method. If you want to display more than five graphs at a time, you should call the method with an appropriate parameter.

Before you can make any of the graphing methods work, you have to set x and y ranges (or, in general, the ranges for your horizontal and vertical variables), in order for your instance of GraphingBoard to know how to translate functional values to pixel values and vice versa. Once you decided on the ranges for x and y (based on the user’s input or your own assignment), you have to call the method:

- **`instance.setVarsRanges(a:Number,b:Number,c:Number,d:Number): void`**

Default: no range is set until you call the method.

In *Function Grapher in ActionScript 3.0 – Template 1*, we want to open with the standard range -10, 10, -10, 10. Hence, we call for our instance of GraphingBoard named “board”:


```
board.setVarsRanges(-10,10,-10,10);
```

Later, within the function `makeGraphs` after we parse the user's input for x and y ranges and assign values to the range variables `xmin`, `xmax`, `ymin`, `ymax`, we call:

```
board.setVarsRanges(xmin,xmax,ymin,ymax);
```

Once the variables ranges are set, we can draw the horizontal and the vertical axes via the method:

- **`instance.drawAxes(): void`**

For example in any of the templates we use:

```
board.drawAxes();
```

Graphs of functions or curves are drawn using the method:

- **`instance.drawGraph(num:int, thick:Number, aVals:Array, c:Number):Array`**

The method takes four parameters. The first, an integer, is the number of the graph being drawn. This integer must not exceed the maximum number of graphs to be displayed at one time (set by `instance.setMaxNumGraphs(..)` method. The integer will also determine the depth of the graph being drawn in the internal stacking order of your instance of `GraphingBoard`. (A graph with a higher number will appear in front of a graph with lower number.) The second parameter will determine the thickness of your graph, in pixels. The third parameter should be an array whose elements are two-element arrays. Each of the two-element arrays represents the x and y coordinates of a point within (or outside) the graphing board. These coordinates are in functional terms; they will be translated to pixel values by the method. For the method to work properly, this array, denoted above by `aVals`, should consist of consecutive points along a graph or a curve which will be joined by lineal elements to form the actual graph. (Although, the method will join the consecutive points in the array by lineal elements regardless what the points represent.) The last parameter is responsible for the color of the graph. The value for a color should be passed in hex.

Below we illustrate how the method is used in *Function Grapher in ActionScript 3.0 – Template 1*. The portion of the script below comes from within `makeGraphs` function. Earlier within the function, the x and y ranges, `xmin`, `xmax`, `ymin`, `ymax`, have been already determined by parsing the user's input. A string variable, `sFunction1`, and a `CompiledObject` variable, `compObj1`, have been declared already.

```
board.setVarsRanges(xmin,xmax,ymin,ymax);

board.drawAxes();

/*
Determining the value of xstep used later for creating an array
of points to be plotted.
*/
```

```

xstep=(xmax-xmin)/points;

/*
We are retrieving the formulas that our user
entered for the first function to be graphed.
*/

sFunction1=InputBox1.text;
.....

/*
We will compile the formula using MathParser,
if the user entered it.
*/

if(sFunction1.length>0){

/*
Evoking our MathParser "doCompile" method
to compile the first formula entered by the user.
Recall that the instance of MathParser created above is called
procFun. If an error is found during compiling,
a message is sent to board.ErrorBox and the function quits.
*/

compObj1=procFun.doCompile(sFunction1);

if(compObj1.errorStatus==1){

    board.ErrorBox.visible=true;

    board.ErrorBox.text="Error in f1(x). "+compObj1.errorMes;

    return;

}

/*
If no error is found we create an array of points, flArray, to be
plotted. Each entry of the array consists of a pair of [x,y] values.
x values are determined by starting from xmin and adding step-by-step
the value which brings us to the next point on the x axis.
This value is xstep and depends on the number of points chosen above.
To obtain to corresponding values of y, we evoke the procFun.doEval
method.
*/

for(i=0;i<=points;i++){

f1Array[i]=[xmin+i*xstep,procFun.doEval(compObj1.PolishArray,[xmin+xstep*i]);

}

/*
Observe, that the array of points created above contains
functional values. Those values will be converted to pixel values
by "board". board.drawGraph method evoked below will do it for

```

```

        us.
        */

board.drawGraph(1,1,f1Array,0xFF0000);

}

```

The drawGraph method returns an array. If the tracing cursor is “cross” (the default), the returned array is the original aVals array which was passed to the method with all x and y coordinates of all points translated to their pixel equivalents. Being able to retrieve this translated array in your applet is valuable for tracing purposes. The returned array gives you the consecutive positions for the cross cursor when tracing the graph. If you do not have a tracing mechanism in your applet, you can ignore the array returned by the method. In *Function Grapher in ActionScript 3.0 – Template 2*, we use the returned array to login positions for tracing. Note the following lines within makeGraphs function:

```

f1PixArray=board.drawGraph(1,1,f1FunArray,0xFF0000);

f2PixArray=board.drawGraph(2,1,f2FunArray,0x0000FF);

```

If the cursor is “arrow”, drawGraph returns an array of three-element arrays. Each of the three-element arrays gives you a position of the tracing arrow along the curve as well as the arrow’s rotation. Again, the returned array serves a possible tracing mechanism. In our template *Parametric Curves in ActionScript 3.0 – Basic*, we used the returned array. Note the line in drawCurve function:

```

arrowPos=board.drawGraph(1,1,fArray,0xFF0000);

```

“arrowPos” is a global array variable used later to trace a curve with the arrow cursor.

Public Methods – Clearing the Graphing Board

To clear the graphs you use the method:

- **`instance.cleanBoard(): void`**

The method erases all graphs, the x and y axes, and resets the x and y ranges to undefined. The method does not erase the user’s drawing. The latter is accomplished by

- **`instance.eraseUserDraw(): void`**

We separated these two methods since, typically, you want the GRAPH button to clean the graphing board but not to erase the user’s drawing. (The user may possibly be experimenting with drawing functions.)

Public Methods – Other

Here are some other possibly useful methods of GraphingBoard class.

- **`instance.getMaxNumGraphs(): int`**

Use this method if you are not sure what the maximum number of graphs is set to. Similarly:

- **`instance.getBoardWidth(): Number`**
- **`instance.getBoardHeight(): Number`**
- **`instance.getVarsRanges(): Array`**

The next four methods allow you to convert functional coordinates to their pixel equivalents and vice versa. Note: these methods will work only if the ranges for your horizontal and vertical variables are set. They will work correctly regardless what names you gave your horizontal and vertical variables.

- **`instance.xtoPix(a:Number): Number`**
- **`instance.ytoPix(a:Number): Number`**
- **`instance.xtoFun(a:Number): Number`**
- **`instance.ytoFun(a:Number): Number`**

A couple of testing methods:

- **`instance.isLegal(a:*) : Boolean`**

The method returns “true” if “a” is of the numerical datatype and it is a finite number. Otherwise, the method returns “false”.

- **`instance.isDrawable(a:*) : Boolean`**

The method returns “true” if “a” is of the numerical datatype and it is a finite number, and its absolute value does not exceed 5000. Otherwise, the method returns “false”. The reason you may want to have a test of such kind is that an attempt to draw an object, a portion of a graph or a cursor, located very far away from the graphing board (in pixels), you may encounter unexpected results. In our templates, you can draw outside the graphing board as the board is masked. But you shouldn’t draw objects which are too far away.

Finally, if you want to remove an instance of GraphingBoard at runtime, you should call

- **`instance.destroy(): void`**

The method removes all listeners set by your instance of GraphingBoard, clears all drawings, and sets all the Sprites created by the instance to null.

Public Properties

The only public property of GraphingBoard (except for those inherited from Sprite) is

- ***instance.ErrorMessage***

It is a dynamic text field in which error messages to the user can be displayed. As we described above, you can set the size, the position, and the formatting for the text field by using GraphingBoard methods. You can also apply Flash's TextField class properties and methods to ErrorMessage, e.g.:

```
board.ErrorMessage.visible=true;

board.ErrorMessage.text="Error in f1(x). "+compObj1.errorMessage;
```

bkde.as3.utilities.HorizontalSlider

Description

Flash CS3 has a slider component. Our class provides a light-weight, easily customizable alternative. The class HorizontalSlider extends Sprite. Hence, it inherits from Sprite. Each instance of HorizontalSlider consists of a track, 3 pixels wide, with four tick marks, and a draggable knob. You can control the length, the style, the colors, and the appearance of the slider using the methods listed below.

Constructor

The constructor is evoked with the word “new” and takes two parameters. The first parameter is the length, in pixels, of the slider to be created. The second, a String parameter, determines the style of the draggable knob:

- ***new HorizontalSlider(len:Number, style:String);***

The available styles for the knob are” “triangle” and “rectangle”.

In our template *Parametric Curves in ActionScript 3.0 – Basic*, we create a slider that is 250 pixels long and has a triangular knob:

```
var hsSlider:HorizontalSlider=new HorizontalSlider(250,"triangle");

addChild(hsSlider);

hsSlider.x=30;
```

```
hsSlider.y=435;
```

We store our instance of HorizontalSlider in a variable called “hsSlider” whose datatype is HorizontalSlider. As with all display objects, we have to add our slider to be a child of an object already on the Display List, in this case to the main movie. We set the position of our slider using properties inherited from Sprite.

Public Methods

Most methods of the class are meant to give you control over your slider’s visual attributes. The names of the properties are quite self-explanatory.

- ***instance.changeKnobColor(c:Number) : void***

Default: dark gray.

The color should be passed in hex, as in *Parametric Curves in ActionScript 3.0 – Basic* :

```
hsSlider.changeKnobColor(0xCC0000);
```

- ***instance.changeKnobSize(c:Number) : void***

Default: 8 (in pixels).

- ***instance.changeKnobOpacity(n:Number) : void***

Default: 1.0 – completely opaque.

You may find this method useful with a rectangular knob if you want the tick marks underneath to show.

- ***instance.changeKnobLeftLine(c:Number) : void***
- ***instance.changeKnobRightLine(c:Number) : void***

Default: white, black.

The methods change the colors of the left and the right outline of the knob to create a 3D effect with your coloring scheme.

- ***instance.changeTrackOutColor(c:Number) : void***
- ***instance.changeTrackInColor(c:Number) : void***

Default: dark gray, white.

The methods change the colors of the outline of the track and the line inside to match your coloring scheme.

- **`instance.setKnobPos(p:Number): void`**

The method adjusts the x coordinate of the knob along the horizontal track. 0 corresponds to the left end of the slider. The method is usually used to set the initial position of the knob.

- **`instance.getKnobPos(): Number`**

This extremely important method allows you to make your applet respond to the changing horizontal position of the knob as the user drags the knob along the track. Here is an excerpt from *Parametric Curves in ActionScript 3.0 – Basic*. (The function `KnobPosToFun` translates the horizontal position of the knob in pixels to the equivalent value of the parameter “t”. The function has been defined earlier in the script. The same is true for the function `getArrowPos`. `KnobBox` is a dynamic text field residing on the Stage in which the current value of “t” is displayed as the knob moves.) The fragment of code below uses the property of `HorizontalSlider`, `isPressed` discussed in Properties section.

```
/*
The event listener is assigned to the Stage which listens to the
movement of the mouse and responds to it provided the slider's knob
is pressed. This event listener causes the arrow to move along
a curve.
*/

stage.addEventListener(MouseEvent.CLICK, handleMove);

function handleMove(e:MouseEvent):void {

    if (hsSlider.isPressed) {

        var curKnobPos:Number;

        var curArrowPos:Array;

        curKnobPos=KnobPosToFun(hsSlider.getKnobPos());

        if (isTRangeSet) {

            if (isGraphToTrace) {

                curArrowPos=getArrowPos(curKnobPos);

                board.setArrowPos(curArrowPos[0], curArrowPos[1], curArrowPos[2]);

                board.arrowVisible(true);

            }

            KnobBox.text=String(Math.round(curKnobPos*1000)/1000);

        } else {

            KnobBox.text="";

        }

    }

}
```

```

        }

        e.updateAfterEvent();
    }

}

```

Here are the last two methods of lesser importance.

- **`instance.getSliderLen() : Number`**

The method returns the slider's length.

Finally, if you want to remove an instance of `HorizontalSlider` at runtime, you should call

- **`instance.destroy() : void`**

The method removes all listeners set by your instance of `HorizontalSlider`, clears all drawings, and sets all the Sprites created by the instance to null.

Public Properties

Each instance of `HorizontalSlider` has one public property (except for those inherited from `Sprite`). This property, “isPressed”, is a **read-only** property:

- **`instance.isPressed: Boolean`**

The property is set by the class to “true” if the user presses the mouse button over the knob. The property is reset to the default – “false” when the user releases the mouse button.

bkde.as3.utilities.VerticalSlider

Description

The class is nearly identical to `HorizontalSlider` class. It has the same methods and properties. The only difference is that the track is drawn vertically and all references in the description above to the x (or horizontal) coordinate should be replaced by references to the y (or vertical) coordinate. Since `HorizontalSlider` inherits from `Sprite`, any instance of `HorizontalSlider` can be positioned vertically via `Sprite.rotation` property. The reason we have a separate class has to do with slight differences in the way the vertical slider is drawn which give it a more pleasing appearance.

Constructor

The constructor is evoked with the word “new” and takes two parameters. The first parameter is the length, in pixels, of the slider to be created. The second, a String parameter, determines the style of the draggable knob:

- `new VerticalSlider(len:Number, style:String);`

The available styles for the knob are” “triangle” and “rectangle”.

Public Methods

Same as for HorizontalSlider.

Public Properties

Same as for HorizontalSlider.

July 23, 2007